

A 2-D VHDL-AMS Model for Disk-Shape Piezoelectric Transducers

Jean-Marc Gallière
University of Montpellier 2
Polytech'Montpellier, France
galliere@polytech.univ-montp2.fr

Philippe Papet
University of Montpellier 2
LPMC, France
papet@polytech.univ-montp2.fr

Laurent Latorre
University of Montpellier 2
LIRMM, France
latorre@lirimm.fr

ABSTRACT

Piezoelectric materials are widely used for many applications such as sensors, actuators. Today, their integration in microelectronics processes like CMOS [1] requires the development of advanced realistic behavioral models. Until now, these models were limited to only one ceramic's operation mode, i.e., thickness or planar. Moreover, the robustness of piezo-electronic system cannot be adequately addressed as long as models are not improved, in particular by taking into account further real phenomena. This article proposes to merge, in a new behavioral model, the two operation modes. It is demonstrated that the electrical behavior of the proposed model is in very good agreement with the real ceramic behavior.

1. INTRODUCTION

In today worldwide competition, the product development cycle is an issue for the design engineer. During the design process of a new product, an important effort must be achieved by the designer to decrease in a drastic way the time to market. In a short time, the engineer has not only to validate design specifications but also to ensure the product's robustness. In this context, it becomes necessary to implement as soon as possible in the design process faithful model of the entire component involved in the product. For several decades, the industry of piezoelectricity has attempted to create such models. Nevertheless, the modeling of piezoelectric ceramics, which are component located between two worlds - mechanical and electrical - is not an easy task.

An important issue when designing ultrasonic based systems is the knowledge of the ceramic behavior in both the mechanical and the electrical domain. Indeed, by definition the two fields closely interact. The electromechanical interaction, represented by electrical equivalent circuits, was first introduced by Mason [2]. Redwood [3] enhanced this electromechanical model by incorporating a transmission line, making possible to extract useful information on the temporal response of the piezoelectric component. Thus, it is possible to represent the propagation time for a mechanical wave from one side of the ceramic to the other.

The piezoelectric crystal deforms in different ways at different frequencies. Those various deformations are called the vibration modes. Like most solid bodies, the vibration modes result from a system of standing waves. These modes can be expressed from a wave equation, in association with a series of overtone modes which are solutions of the same set of equations. A number of research works have been conducted in the past years dealing with the ceramic's behavior. Although these models perfectly match the electrical characteristics of the piezoelectric transducers, they suffer from a strong limitation: they cannot implement several vibration modes simultaneously. Recently, in [4] a new unified model was proposed. This model implements two vibrating mode in the same SPICE model.

To carry on this study, the objective of this paper is to present a unified behavioral model permitting to handle together the planar and thickness bulk vibration modes of ceramic disks. For this study, the model is implemented with the VHDL-AMS behavioral language. We prefer the use of VHDL-AMS because it provides powerful capabilities for modeling components and their interactions in multiple energy domains. To perform the implementation, the SystemVision™ [5] tool from Mentor Graphics® is used. SystemVision™ is an intuitive virtual-prototyping environment. This environment provides multi-level model integration required for true systems design and analysis.

The remainder of the paper is organized as follows. Section 2 recalls how the thickness vibration model is implemented in the literature. The third section introduces the new unified model that we propose. Section 4 compares simulation results obtained with our model with real ceramic measurements. Finally, we conclude in section 5.

2. THICKNESS BEHAVIORAL MODELS

This study will be limited to the cases of ceramic disks (probably the most convenient shape to fabricate) polarized (P) along the 3-axis (it is conventional to align the coordinate system with the poling directions) which is the axis of applied electric field (E). As a consequence the crystalline symmetry of the poled polycrystalline ceramics, which have ∞ -fold symmetry in a plane normal to the poling direction belongs to 6mm group in the hexagonal symmetry system. Therefore, for the analysis, a cylindrical coordinate system

with its origin located at the center of the disc is most suitable. Due to the symmetry, only thickness and radial (planar) modes are excited (Fig. 1) and axes r and z are assumed to be pure mode propagation directions.




Fig. 1: Typical vibration modes of ceramic disks

Moreover, since biased surfaces are the two parallel surfaces of the disc, only the component E_z of the exciting electric field has to be considered. Taking into account these assumptions, a 2-D analytical model of piezoceramic disk has been developed in [6].

2.1. Electrical study

From the equation of acoustic wave's propagation in piezoelectric materials, it is possible to write linear relations linking, on the one hand the mechanical magnitudes (force F and speed of particles u) which are preserved at an interface and, on the other hand electrical quantities (applied potential v_3 and intensity i_3 of the current). Having an input vector of dimension 3 (i.e., two mechanicals and one electrical input) leads to an impedance matrix:

$$\begin{bmatrix} F1 \\ F2 \\ v3 \end{bmatrix} = -j \cdot \begin{bmatrix} Z/\tan\left(\frac{\omega d}{u}\right) & Z/\sin\left(\frac{\omega d}{u}\right) & h/\omega \\ Z/\sin\left(\frac{\omega d}{u}\right) & Z/\tan\left(\frac{\omega d}{u}\right) & h/\omega \\ h/\omega & h/\omega & 1/\omega C0 \end{bmatrix} \cdot \begin{bmatrix} u1 \\ u2 \\ i3 \end{bmatrix}$$

Where $F1$ and $F2$ (N) symbolize the forces, u_1 and u_2 (m/s) are the particle velocities inside the material, ω the angular frequency (rad/s) and $h_{33} = e_{33}/\epsilon^S$ (V/m) the piezoelectric constant with e_{33} (C/m²) the piezoelectric coefficient. The mechanical impedance Z (rayl) is calculated knowing the ceramic density ρ (kg/m³), the particle velocity u (m/s) and the area A (m²) by using $Z = \rho \cdot u \cdot A$. The equivalent circuit of Fig. 2 can be easily derived from the previous piezoelectric impedance matrix [7].

The diagram of Fig. 2 explains the port definition for a thickness-mode transducer along with Redwood's version of Masson's equivalent circuit. The model consists of a capacitance $C0$, a negative capacitance $-C0$, an ideal transformer and a transmission line. $C0$ is the so called piezoceramic clamped capacity:

$$C0 = \epsilon^S \cdot \frac{A}{d}$$

Where ϵ^S (C²/Nm²) is the ceramic permittivity with zero or constant strain, A (m²) is the area electrodes and d (m) his thickness.




Fig. 2: Transducer and his equivalent circuit of Mason as adapted by Redwood

The mechanical part of the piezoelectric transducer is easily represented using a transmission line model. This class of component is well known and modeled; in addition, it fits perfectly in this context. In first approximation, two parameters are sufficient to entirely define the mechanical part of the transducer, i.e., the impedance Z and the sound propagation delay td through the transducer.

2.2. Thickness-Mode VHDL-AMS Behavioral model

A VHDL-AMS thickness-mode model was developed in the past [8]. This model (Fig. 3) is a direct transcription of the Redwood's model [3]. Both, the two resistances r_f and r_b represent the acoustic impedance for respectively the front and the back of the transducer.




Fig. 3: Equivalent circuit of Redwood's model

The VHDL-AMS implementation of the previous model (Fig. 3) is divided in two parts (Listing 1). First is the declaration of the *entity* which is composed of the physical characteristics of the transducer and the different *terminals* used to connect the electric input. The second part of the model is the *architecture* which establishes the physic laws related to the mathematical relation between each terminal.

```

Listing 1 : Thickness Redwood model
library ieee;
use ieee.math_real.all;
use ieee.electrical_systems.all;
entity redwood is
generic (
    C0 : real := 1.24e-9;
    kt : real := 2.95;
    Z0 : real := 7009.0;
    td : real := 2.20e-7);
port (terminal p, m : electrical);
end entity redwood;
architecture bhv of redwood is
terminal p1, t1, t22, t11, km : electrical;

```

```

quantity v1    across i1 through p to m;
quantity v2    across i2 through p to p1;
quantity vte   across ite through p1 to m;
quantity pti   across uti through t1 to km;
begin
  i1    == C0 * v1'.dot;
  i2    == -C0 * v2'.dot;
  pti   == kt * vte;
  uti   == ite/kt;
  ceramic : entity work.acousticlayer
    generic map (Z0=>Z0, td=>td)
    port map (p1=>t11,m1=>km,p2=>t22,m2=>km);
  rf : entity WORK.resistance
    generic map (rnom=>0.08)
    port map (plus=>t11 , moins=>t1 );
  rb : entity WORK.resistance
    generic map (rnom=>0.08)
    port map (plus=>t22 , moins=>t1 );
end architecture bhv;

```

The mechanical part, called the acoustic layer, is described in listing 2. This model corresponds to the electric equivalent circuit of Branin [9]. The entity is composed of two acoustic port connections. One is connected to the propagation medium at the front side i.e. rf and the other is attached to the propagation medium at the back side i.e. rb. In our case the ceramic is immersed in air. Z_0 and td represent respectively the transducer's impedance and the sound transit time across the transceiver.

Listing 2 : Acoustic layer model

```

library ieee;
use ieee.math_real.all;
use ieee.electrical_systems.all;

entity acousticlayer is
generic ( Z0, td : real );
port ( terminal p1, m1, p2, m2 : electrical);
end entity acousticlayer;

architecture bhva of acousticlayer is
terminal t111, t222 : electrical;

quantity fi    across p1 to m1;
quantity ft    across p2 to m2;
quantity fii   across uiz through t111 to m1;
quantity fiz   across ui through t111 to p1;
quantity ftz   across ut through t222 to p2;
quantity ftt   across utz through t222 to m2;

begin
  ftt == fi'DELAYED(td) - ftz;
  fii == ft'DELAYED(td) - fiz;
  fiz == (uiz + utz'DELAYED(td))*Z0/2.0;
  ftz == (utz + uiz'DELAYED(td))*Z0/2.0;
end architecture bhva;

```

In Fig. 4, a comparison is made between the electrical impedance $Z = v_1/(i_1 + i_2)$ obtained with the behavioral model by simulation and a ceramic measurement performed with an Agilent 4294A precision impedance analyzer [10]. The Ferroperm [11] commercial transducer used in this study is a circular PZ26 of 16mm diameter by 1mm thick. The experimental set-up is composed of this analyzer with his impedance test kit and a spring-clip fixture which applies

very little mechanical loading in such a way that the sample is under free piezoelectric resonator condition.




Fig 4: Impedance comparison between experiment and simulation for Thickness-mode Behavioral model

As one can observe in Fig. 4, the simulated harmonic resonances appear at the desired frequency, i.e., ~2MHz. However, since model does not take into account the two kinds of losses (mechanical and dielectric), the peaks appear sharper in the simulation. This can be corrected by modeling the losses starting from the model of the transmission line as explain by Püttmer et al. in [12]. Moreover, the small difference found for the first thickness harmonic arises from the fact that the two electrical Ni/Au contact layers are not considered by the model. To this end, the model can be easily enhanced with the adjunction of two appropriated transmission lines instead of the conducting layers.

For lower frequencies (<2MHz), the simulated curve differs more significantly from the experimental one. In fact, the peaks which emerge in the measured curve embody the planar resonance not managed by the behavioral model.

From Fig. 4, it appears that mismatches exist between real measurement and simulation. Actually, the models are valid only for a limited part of the frequency range. But what happen if the two modes are mixed together? It could be the case when ceramics' thicknesses increase regarding the diameter. For this reason, in the next section a new model is presented. This model will be able to manage, together, the two modes of operation on the entire frequency spectrum of ceramic's use.

3. THE NEW UNIFIED MODEL

Usually, the two modes of vibration are independently considered in distinct models, each one of them treating only a part of the real component operation. The only way to simulate the real ceramic's behavior is to integrate in a one and only model both the planar mode and the thickness mode. The proposed model must be able to take into account both the coupling between the planar and thickness modes and the mechanical interactions of the disk with the media on major faces and on curved surface. To this end, Iula et al. [6] demonstrate that the current i across a disk ceramic is expressed by the equation (1):

$$i = j\omega.C0.v_1 - ktT.(u_1 + u_2) - ktP.(u_3 + u_4) \quad (1)$$

Where: v_1 is the potential through the transducer and

$$u_{1,2} = \sqrt{\frac{C_{33}^D}{\rho}} ; u_{3,4} = \sqrt{\frac{C_{11}^D}{\rho}} ; ktT=h_{33}.C0 ; ktP=h_{31}.C0$$

Equation (1) along with the Kirchoff's law on currents leads to the equivalent electrical representation located on the left side of Fig. 5. Note that the current $i_{te} = ktT.(u_1+u_2)$ and the current $i_{tP} = ktP.(u_3+u_4)$.

The overall current i which is flowing across the ceramic's electrodes is related to the particle velocity as described in (1) by u_1, u_2, u_3 and u_4 . The force applied on each ceramic's surface is linked to the current by the equation $f=h.i/j\omega$ (2) where h , the piezoelectric constant, takes the value of h_{33} or h_{31} depending of which mode is involved. In the case of thickness mode, substituting equation (1) in (2) gives:

$$f_{thickness} = ktT \left[v_1 - \frac{1}{j\omega C0} . ktT . (u_1 + u_2) - \frac{1}{j\omega C0} . ktP . (u_3 + u_4) \right] \quad (3)$$

Likewise, for the planar mode the force can be expressed by:

$$f_{planar} = ktP \left[v_1 - \frac{1}{j\omega C0} . ktP . (u_3 + u_4) - \frac{1}{j\omega C0} . ktT . (u_1 + u_2) \right] \quad (4)$$

The electro-mechanical part of the model is achieved according to the equations (3) and (4). To complete the model, two transmission lines have to be added to represent the mechanical part of the ceramic (Fig. 5). Each one of it takes into account the acoustic wave propagation according to a privileged direction, planar or in-thickness. Consequently, it is possible to connect 4 independent acoustic forces to each surfaces of the transducer, i.e., F1, F2, F3 and F4.




Fig. 5: The unified electrical model

In listing 3, the electrical model of Fig. 5 is implemented in VHDL-AMS.

Listing 3 : UGP Behavioral model

```

library ieee;
use ieee.math_real.all;
use ieee.electrical_systems.all;

entity ugp is
generic (
    C0 : real := 1.24e-9;
    Z0 : real := 7009.0;
    ktP : real := -0.56;
    tdP : real := 3.53e-6;
    ktT : real := 2.95;
    tdT : real := 2.20e-7);
port ( terminal p, m : electrical);
end entity ugp;

architecture bhvugp of ugp is

terminal p1,t1,t22,t11,km : electrical;
terminal p2,t2,t11P,t22P : electrical;

quantity v1 across i1 through p to m;
quantity v2 across i2 through p to p1;
quantity vte across ite through p1 to m;
quantity pti across uti through t1 to km;

quantity v3 across i3 through p to p2;
quantity vteP across iteP through p2 to m;
quantity ptiP across utiP through t2 to km;

begin
    i1 == C0 * v1'dot;
    i2 == -C0 * v2'dot;
    pti == ktT * (v1-v2-v3);
    uti == ite/ktT;
    i3 == -C0 * v3'dot;
    ptiP == ktP * (v1-v2-v3);
    utiP == iteP/ktP;

    ceramicT : entity work.acousticlayer
    generic map (Z0=>Z0, td=>tdT)
    port map (p1=>t11,m1=>km,p2=>t22,m2=>km);
    rfT : entity WORK.resistance
    generic map (rnom=>0.08)
    port map (plus=>t11 , moins=>t1 );
    rbT : entity WORK.resistance
    generic map (rnom=>0.08)
    port map (plus=>t22 , moins=>t1 );

    ceramicP : entity work.acousticlayer
    generic map (Z0=>Z0, td=>tdP)
    port map (p1=>t11P,m1=>km,p2=>t22P,m2=>km);
    rfP : entity WORK.resistance
    generic map (rnom=>0.08)
    port map (plus=>t11P , moins=>t2 );
    rbP : entity WORK.resistance
    generic map (rnom=>0.08)
    port map (plus=>t22P , moins=>t2 );

end architecture bhvugp;

```

4. EXPERIMENT VERSUS SIMULATION

In order to obtain an experimental validation of the proposed model, we used the same impedance measurement setup than the one previously described for the thickness mode. Fig. 6 shows the measured and simulated amplitude of the input impedance for a 2mm thick ceramic disk. A wide

frequency spectrum is chosen to demonstrate the good agreement between the model and the dual intrinsic resonance of the ceramic. For high frequencies a divergence is observed between measurement and simulation. This slight difference comes from the measurement setup which is not considered into the simulation.




Fig. 6: Impedance comparison between experiment and simulation for the unified behavioral model (2mm thick)

In the same way, Fig. 7 shows the similar comparison for a thinner ceramic disk. Here also a good agreement is observed (better for the first planar-mode than for the first thickness-mode resonance frequency).




Fig. 7: Impedance comparison between experiment and simulation for the unified behavioral model (1mm thick)

5. CONCLUSION

Based on the analysis of the electromechanical behavior of the PZT ceramic we have presented a comprehensive method for the development of a new behavioral model of cylinder shaped piezoceramic elements. This model has been validated for various diameter-to-thickness ratios. It describes the electromechanical coupling between thickness and planar modes by coupling electrically and mechanically in explicit form the two dimensional vibration. In order to validate this unified behavioral model and to determine the accuracy of results, an experimental validation of the model has been carried out. Calculated impedance versus frequency is then successfully compared with measured values.

REFERENCES

- [1] P. H. Sung et al., "The Method for Integrating FBAR with Circuitry on CMOS Chip," IEEE International Frequency Control Symposium and Exposition, 23-27 August 2004, pp. 562-565
- [2] W. P. Mason, "Electromechanical Transducers and Wave Filters," Princeton, NJ: Van Nostrand, 1948, pp. 201-209, 399-404
- [3] M. Redwood, "Transient Performance of a Piezoelectric Transducer," J. Acoust. Soc. Amer., vol. 33, no 4, April 1961, pp527-536
- [4] J. M. Galliere et al., "A Unified Electrical SPICE Model for Piezoelectric Transducers," IEEE International Behavioral Modeling and Simulation Workshop, September 20-21, 2007, pp138-142
- [5] <http://www.mentor.com/SystemVision>
- [6] A. Iula, N. Lamberti and M. Pappalardo, "An Approximated 3-D Model of Cylinder-Shaped Piezoceramic Element for Transducer Design," IEEE Trans. On Ultrasonics, Ferroelectrics and Frequency Control, vol. 45 no 4, July 1998, pp1056-1064
- [7] W. P. Mason, "Physical Acoustics," New York, Academic Press, vol. 1 part A, 1964
- [8] R. Guelaz et al., "Double element ultrasonic piezoceramic transducer modeling with VHDL-AMS: application to B/A nonlinear ultrasonic parameter measurement in pulse-echo mode," Electronic Journal, Technical Acoustics, vol. 10, May 2005
- [9] F. Branin, "Transient Analysis of Lossless Transmission Lines," in proceeding of IEEE, v. 55, 1967, pp2012-2013
- [10] <http://cp.literature.agilent.com/litweb/pdf/5968-3809E.pdf>
- [11] <http://www.ferroperm-piezo.com>
- [12] A. Püttmer et al., "SPICE Model for Lossy Piezoceramic Transducers," IEEE Trans. On Ultrasonics, Ferroelectrics and Frequency Control, vol. 44 no 1, January 1997, pp60-65